```
> restart: # Newsvendor-Incomplete.mw
> theta:=0.5;
```

$$\theta := 0.5 \tag{1}$$

```
> a:=9/10; b:=9/10;
```

$$a := \frac{9}{10}$$

$$b := \frac{9}{10} \tag{2}$$

```
> lambda:=1/30; mu:=1/20;
```

$$\lambda := \frac{1}{30}$$

$$\mu := \frac{1}{20} \tag{3}$$

```
> s:=[15,9]; c1L:=6; c1H:=10; c[2]:=5;
```

$$s := [15, 9]$$

$$c1L := 6$$

$$c1H := 10$$

$$c_2 := 5 \tag{4}$$

```
> f:=x->lambda*exp(-lambda*x);
```

$$f := x \rightarrow \lambda\, e^{-\lambda x} \tag{5}$$

```
> g:=y->mu*exp(-mu*y);
```

$$g := y \rightarrow \mu\, e^{-\mu y} \tag{6}$$

```
> int(f(x),x=0..infinity); int(g(y),y=0..infinity);
```

$$1$$

$$1 \tag{7}$$

```
> B:='((u-x)/b)+v';
```

$$B := \frac{u - x}{b} + v \tag{8}$$

```
> A:='((v-y)/a)+u';
```

$$A := \frac{v - y}{a} + u \tag{9}$$

```
> # J[1]
> T[1,1]:=unapply('s[1]'*Int('x*f(x)',x=0..u),u,v);
```

$$T_{1,1} := (u, v) \rightarrow s_1 \left(\int_0^u x f(x)\, dx \right) \tag{10}$$

```
> value(T[1,1](u,v));
```

$$450 - 450\, e^{-\frac{1}{30} u} - 15\, e^{-\frac{1}{30} u}\, u \tag{11}$$

```
> T[1,2]:=unapply('s[1]'*u*Int('f(x)',x=u..infinity),u,v);
```

$$T_{1,2} := (u, v) \to s_1\, u \left(\int_u^{\infty} f(x)\, dx \right) \tag{12}$$

```
> value(T[1,2](u,v));
```

$$15\, e^{-\frac{1}{30} u}\, u \tag{13}$$

```
> T[1,3]:=unapply('s[1]'*Int(Int('b*(y-v)*g(y)*f(x)',y=v..'B'),x=
  0..u),u,v);
```

$$T_{1,3} := (u, v) \to s_1 \left(\int_0^u \int_v^B b\, (y - v)\, g(y)\, f(x)\, dy\, dx \right) \tag{14}$$

```
> value(T[1,3](u,v));
```

$$270\, e^{-\frac{1}{20} v} + \frac{2835}{2} e^{-\frac{1}{18} u - \frac{1}{20} v} + \frac{45}{2} e^{-\frac{1}{18} u - \frac{1}{20} v} u - \frac{3375}{2} e^{-\frac{1}{20} v - \frac{1}{30} u} \tag{15}$$

```
> T[1,4]:=unapply('s[1]'*Int(Int('(u-x)*g(y)*f(x)',y='B..
  infinity'),x=0..u),u,v);
```

$$T_{1,4} := (u, v) \to s_1 \left(\int_0^u \int_B^{\infty} (u - x)\, g(y)\, f(x)\, dy\, dx \right) \tag{16}$$

```
> value(T[1,4](u,v));
```

$$-\frac{45}{2} e^{-\frac{1}{18} u - \frac{1}{20} v} u - \frac{2025}{2} e^{-\frac{1}{18} u - \frac{1}{20} v} + \frac{2025}{2} e^{-\frac{1}{20} v - \frac{1}{30} u} \tag{17}$$

```
> #T[1,5]:=(u,v)->-c[1]*u;
> JL[1]:=unapply(add(value(T[1,i](u,v)),i=1..4)-c1L*u,u,v);
```

$$JL_1 := (u, v) \to 450 - 450\, e^{-\frac{1}{30} u} + 270\, e^{-\frac{1}{20} v} + 405\, e^{-\frac{1}{18} u - \frac{1}{20} v} - 675\, e^{-\frac{1}{20} v - \frac{1}{30} u} - 6\, u \tag{18}$$

```
> JL[1](u,v);
```

$$450 - 450\, e^{-\frac{1}{30} u} + 270\, e^{-\frac{1}{20} v} + 405\, e^{-\frac{1}{18} u - \frac{1}{20} v} - 675\, e^{-\frac{1}{20} v - \frac{1}{30} u} - 6\, u \tag{19}$$

```
> JL[1,u]:=diff(JL[1](u,v),u);
```

$$JL_{1,u} := 15\, e^{-\frac{1}{30} u} - \frac{45}{2} e^{-\frac{1}{18} u - \frac{1}{20} v} + \frac{45}{2} e^{-\frac{1}{20} v - \frac{1}{30} u} - 6 \tag{20}$$

```
> IIL[1]:=unapply(diff(JL[1](u,v),u),u,v);
```

$$IIL_1 := (u, v) \to 15\, e^{-\frac{1}{30} u} - \frac{45}{2} e^{-\frac{1}{18} u - \frac{1}{20} v} + \frac{45}{2} e^{-\frac{1}{20} v - \frac{1}{30} u} - 6 \tag{21}$$

```
> JH[1]:=unapply(add(value(T[1,i](u,v)),i=1..4)-c1H*u,u,v);
```

$$JH_1 := (u, v) \to 450 - 450\, e^{-\frac{1}{30} u} + 270\, e^{-\frac{1}{20} v} + 405\, e^{-\frac{1}{18} u - \frac{1}{20} v} - 675\, e^{-\frac{1}{20} v - \frac{1}{30} u} - 10\, u \tag{22}$$

```
> JH[1](u,v);
```

$$450 - 450\,e^{-\frac{1}{30}u} + 270\,e^{-\frac{1}{20}v} + 405\,e^{-\frac{1}{18}u-\frac{1}{20}v} - 675\,e^{-\frac{1}{20}v-\frac{1}{30}u} - 10\,u \quad (23)$$

```
> JH[1,u]:=diff(JH[1](u,v),u);
```

$$JH_{1,u} := 15\,e^{-\frac{1}{30}u} - \frac{45}{2}\,e^{-\frac{1}{18}u-\frac{1}{20}v} + \frac{45}{2}\,e^{-\frac{1}{20}v-\frac{1}{30}u} - 10 \quad (24)$$

```
> IIH[1]:=unapply(diff(JH[1](u,v),u),u,v);
```

$$IIH_1 := (u,v) \rightarrow 15\,e^{-\frac{1}{30}u} - \frac{45}{2}\,e^{-\frac{1}{18}u-\frac{1}{20}v} + \frac{45}{2}\,e^{-\frac{1}{20}v-\frac{1}{30}u} - 10 \quad (25)$$

```
> # J[2]
> T[2,1]:=unapply('s[2]'*Int('y*g(y)',y=0..v),u,v);
```

$$T_{2,1} := (u,v) \rightarrow s_2 \left(\int_0^v y\,g(y)\,dy \right) \quad (26)$$

```
> value(T[2,1](u,v));
```

$$180 - 180\,e^{-\frac{1}{20}v} - 9\,e^{-\frac{1}{20}v}\,v \quad (27)$$

```
> T[2,2]:=unapply('s[2]'*v*Int('g(y)',y=v..infinity),u,v);
```

$$T_{2,2} := (u,v) \rightarrow s_2\,v \left(\int_v^\infty g(y)\,dy \right) \quad (28)$$

```
> value(T[2,2](u,v));
```

$$9\,e^{-\frac{1}{20}v}\,v \quad (29)$$

```
> T[2,3]:=unapply('s[2]'*Int(Int('a*(x-u)*f(x)*g(y)',x=u..'A'),y=
  0..v),u,v);
```

$$T_{2,3} := (u,v) \rightarrow s_2 \left(\int_0^v \int_u^A a\,(-u+x)\,f(x)\,g(y)\,dx\,dy \right) \quad (30)$$

```
> value(T[2,3](u,v));
```

$$243\,e^{-\frac{1}{30}u} + \frac{85293}{49}\,e^{-\frac{1}{27}v-\frac{1}{30}u} - \frac{243}{7}\,e^{-\frac{1}{27}v-\frac{1}{30}u}\,v - \frac{97200}{49}\,e^{-\frac{1}{20}v-\frac{1}{30}u} \quad (31)$$

```
> T[2,4]:=unapply('s[2]'*Int(Int('(v-y)*f(x)*g(y)',x='A..
  infinity'),y=0..v),u,v);
```

$$T_{2,4} := (u,v) \rightarrow s_2 \left(\int_0^v \int_A^\infty (v-y)\,f(x)\,g(y)\,dx\,dy \right) \quad (32)$$

```
> value(T[2,4](u,v));
```

$$\frac{243}{7}\,e^{-\frac{1}{27}v-\frac{1}{30}u}\,v - \frac{131220}{49}\,e^{-\frac{1}{27}v-\frac{1}{30}u} + \frac{131220}{49}\,e^{-\frac{1}{20}v-\frac{1}{30}u} \quad (33)$$

```
> T[2,5]:=(u,v)->-c[2]*v;
```

$$T_{2,5} := (u,v) \rightarrow -c_2\,v \quad (34)$$

```
> J[2]:=unapply(add(value(T[2,i](u,v)),i=1..5),u,v);
```

$$J_2 := (u, v) \rightarrow 180 - 180\, e^{-\frac{1}{20} v} + 243\, e^{-\frac{1}{30} u} - \frac{6561}{7} e^{-\frac{1}{27} v - \frac{1}{30} u} + \frac{4860}{7} e^{-\frac{1}{20} v - \frac{1}{30} u} - 5\, v \quad (35)$$

```
> J[2](u,v);
```

$$180 - 180\, e^{-\frac{1}{20} v} + 243\, e^{-\frac{1}{30} u} - \frac{6561}{7} e^{-\frac{1}{27} v - \frac{1}{30} u} + \frac{4860}{7} e^{-\frac{1}{20} v - \frac{1}{30} u} - 5\, v \quad (36)$$

```
> J[2,v]:=diff(J[2](u,v),v);
```

$$J_{2,v} := 9\, e^{-\frac{1}{20} v} + \frac{243}{7} e^{-\frac{1}{27} v - \frac{1}{30} u} - \frac{243}{7} e^{-\frac{1}{20} v - \frac{1}{30} u} - 5 \quad (37)$$

```
> II[2]:=unapply(diff(J[2](u,v),v),u,v);
```

$$II_2 := (u, v) \rightarrow 9\, e^{-\frac{1}{20} v} + \frac{243}{7} e^{-\frac{1}{27} v - \frac{1}{30} u} - \frac{243}{7} e^{-\frac{1}{20} v - \frac{1}{30} u} - 5 \quad (38)$$

```
> II[2](30.,40.);
```

$$-2.607522312 \quad (39)$$

```
> eq[1]:=IIL[1](uL,v);
```

$$eq_1 := 15\, e^{-\frac{1}{30} uL} - \frac{45}{2} e^{-\frac{1}{18} uL - \frac{1}{20} v} + \frac{45}{2} e^{-\frac{1}{20} v - \frac{1}{30} uL} - 6 \quad (40)$$

```
> eq[2]:=IIH[1](uH,v);
```

$$eq_2 := 15\, e^{-\frac{1}{30} uH} - \frac{45}{2} e^{-\frac{1}{18} uH - \frac{1}{20} v} + \frac{45}{2} e^{-\frac{1}{20} v - \frac{1}{30} uH} - 10 \quad (41)$$

```
> eq[3]:=theta*II[2](uL,v)+(1-theta)*II[2](uH,v);
```

$$eq_3 := 9.0\, e^{-\frac{1}{20} v} + 17.35714286\, e^{-\frac{1}{27} v - \frac{1}{30} uL} - 17.35714286\, e^{-\frac{1}{20} v - \frac{1}{30} uL} - 5.0 + 17.35714286\, e^{-\frac{1}{27} v - \frac{1}{30} uH} - 17.35714286\, e^{-\frac{1}{20} v - \frac{1}{30} uH} \quad (42)$$

```
> BNESol:=fsolve({eq[1],eq[2],eq[3]},{uL=0..90,uH=0..90,v=0..90});
```

$$BNESol := \{uH = 17.02773201,\ uL = 35.46703659,\ v = 19.75263007\} \quad (43)$$

```
> assign(%);
> uL,uH,v;
```

$$35.46703659,\ 17.02773201,\ 19.75263007 \quad (44)$$

```
> JL[1](uL,v); JH[1](uH,v);
```

$$143.7425863$$
$$41.2377547 \quad (45)$$

```
> theta*J[2](uL,v)+(1-theta)*J[2](uH,v);
```

$$36.30154904 \quad (46)$$

```
> with(plots);
```

$$[animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, \quad (47)$$

conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, densityplot, display, dualaxisplot, fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, intersectplot, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, rootlocus, semilogplot, setcolors, setoptions, setoptions3d, spacecurve, sparsematrixplot, surfdata, textplot, textplot3d, tubeplot]

```
> uL:='uL'; uH:='uH'; v:='v';
```

$$uL := uL$$
$$uH := uH$$
$$v := v \quad \textbf{(48)}$$

```
> implicitplot3d(eq[1],uL=0..90,uH=0..90,v=0..90,axes=boxed);
```




```
> implicitplot3d(eq[2],uL=0..90,uH=0..90,v=0..90,axes=boxed);
```




```
> implicitplot3d(eq[3],uL=0..90,uH=0..90,v=0..90,axes=boxed,
  shading=none);
```




```
> IP1:=implicitplot3d(eq[1],uL=0..90,uH=0..90,v=0..90,shading=
  none):
> IP2:=implicitplot3d(eq[2],uL=0..90,uH=0..90,v=0..90,shading=
  none):
> IP3:=implicitplot3d(eq[3],uL=0..90,uH=0..90,v=0..90):
> display([IP1,IP2,IP3],axes=boxed);
```


v
90
80
70
60
50
40
30
20
10
0
10
20
30
40
50
60
70
80
90
uH
0
10
20
30
40
50
60
70
80
90
uL